Простым условием (отношением) называется выражение, составленное из двух арифметических выражений или двух текстовых величин (иначе их еще называют операндами), связанных одним из знаков:

  < - меньше, чем...
  > - больше, чем...
  <= - меньше, чем... или равно
  >= - больше, чем... или равно
  <> - не равно
  = - равно

Например, простыми отношениями являются следующие:

x-y>10; k<=sqr(c)+abs(a+b); 9<>11; ‘мама’<>‘папа’.

В приведенных примерах первые два отношения включают в себя переменные, поэтому об истинности этих отношений можно судить только при подстановке конкретных значений:

· если х=25, у=3, то отношение x-y>10 будет верным, т.к. 25-3>10

· если х=5, у=30, то отношение x-y>10 будет неверным, т.к. 5-30<10

Проверьте истинность второго отношения при подстановке следующих значений:

1. k=5, a=1, b=-3, c=-8

2. k=65, a=10, b=-3, c=2

Определение. Выражение, о котором после подстановки в него значений переменных можно сказать, истинно (верно) оно или ложно (неверно), называется булевым (логическим) выражением.

Примечание. Название “булевы” произошло от имени математика Джорджа Буля, разработавшего в XIX веке булеву логику и алгебру логики.

Определение. Переменная, которая может принимать одно из двух значений: True (правда) или False (ложь), называется булевой (логической) переменной. Например,

	К:=True;
Flag:=False;
Second:=a+sqr(x)>t


Рассмотрим пример.

Задача. Вычислить значение модуля и квадратного корня из выражения (х-у).

Для решения этой задачи нужны уже знакомые нам стандартные функции нахождения квадратного корня - Sqr и модуля - Abs. Поэтому Вы уже можете записать следующие операторы присваивания:

	Koren:=Sqrt(x-y);
Modul:=Abs(x-y)


В этом случае программа будет иметь вид:

	Program Znachenia;
Uses
  Crt;
Var
  x, y : integer;
  Koren, Modul : real;
Begin
  ClrScr;
  write ('Введите значения переменных х и у через пробел ');
  readln (x, y);
  Koren:=Sqrt(x-y);
  Modul:=Abs(x-y);
  write ('Значение квадратного корня из выражения (х-у) равно ', Koren);
  write ('Значение модуля выражения (х-у) равно ', Modul);
  readln;
End.


Казалось бы, задача решена. Но мы не учли области допустимых значений для нахождения квадратного корня и модуля. Из курса математики Вы должны знать, что можно найти модуль любого числа, а вот значение подкоренного выражения должно быть неотрицательно (больше или равно нулю).

Поэтому наша программа имеет свою допустимую область исходных данных. Найдем эту область. Для этого запишем неравенство х-у>=0, то есть х>=у. Значит, если пользователем нашей программы будут введены такие числа, что при подстановке значение этого неравенства будет равно True, то квадратный корень из выражения (х-у) извлечь можно. А если значение неравенства будет равно False, то выполнение программы закончится аварийно.

Задание. Наберите текст программы. Протестируйте программу со следующими значениями переменных и сделайте вывод.

1. х=23, у=5;

2. х=-5, у=15;

3. х=8, у=8.

Каждая программа, насколько это возможно, должна осуществлять контроль за допустимостью величин, участвующих в вычислениях. Здесь мы сталкиваемся с разветвлением нашего алгоритма в зависимости от условия. Для реализации таких условных переходов в языке Паскаль используют операторы If и Case, а также оператор безусловного перехода Goto.

Рассмотрим оператор If.

Для нашей задачи нужно выполнить следующий алгоритм:

если х>=у,
то вычислить значение квадратного корня,
иначе выдать на экран сообщение об ошибочном введении данных.

Запишем его с помощью оператора If. Это будет выглядеть так.

	if x>=y
  then
    Koren:=Sqr(x-y)
  else
write ('Введены недопустимые значения переменных');


Теперь в зависимости от введенных значений переменных х и у, вычисление квадратного корня может выполняться или не выполняться.

В общем случае полная форма конструкции условного оператора имеет вид:

	if <логическое выражение> 
  then
    <оператор 1> 
  else 
    <оператор 2>


Условный оператор работает по следующему алгоритму.

Сначала вычисляется значение логического выражения, расположенного за служебным словом IF. Если его результат истина, выполняется <оператор 1>, расположенный после слова THEN, а действия после ELSE пропускаются; если результат ложь, то, наоборот, действия после слова THEN пропускаются, а после ELSE выполняется <оператор 2>.

Управляющая структура if может показаться негибкой, так как выполняемые действия могут быть описаны только одним оператором. Иногда может потребоваться выполнение последовательности операторов. В этом случае хотелось бы заключить всю последовательность в воображаемые скобки. В Паскале предусмотрен этот случай.

Если в качестве оператора должна выполниться серия операторов, то они заключаются в операторные скобки begin-end. Конструкция Begin ... End называется составным оператором.

	if <логическое выражение>
  then
    begin 
      оператор 1;
      оператор 2;
      ...
    end
  else 
    begin
      оператор 1;
      оператор 2;
      ...
    end;


Определение. Составной оператор - объединение нескольких операторов в одну группу. Группа операторов внутри составного оператора заключается в операторные скобки (begin-end).

	begin
  оператор 1;
  оператор 2;
end;


С учетом полученных знаний преобразуем нашу программу.

	Program Znachenia;
Uses
  Crt;
Var
  x, y : integer;
  Koren, Modul : real;
Begin
  ClrScr;
  write ('Введите значения переменных х и у через пробел ');
  read (x, y);
  if x>=y
    then
      begin
        Koren:=Sqr(x-y);
        Modul:=Abs(x-y);
        write ('Значение квадратного корня из выражения (х-у) равно ', Koren);
        write ('Значение модуля выражения (х-у) равно ', Modul);
      end
    else
      write ('Введены недопустимые значения переменных');
  readln;
End.


Составным оператором является и такой оператор

	begin
  S:=0;
end.


Cимвол “;” в данном случае разделяет оператор присваивания S:=0 и пустой оператор.

Пустой оператор не влечет никаких действий и в записи программы никак не обозначается.

Например, составной оператор

begin
end.

включает лишь один пустой оператор.

Если Вы обратили внимание, программа на языке Паскаль всегда содержит один составной оператор - раздел операторов программы.

Внимание! Перед служебным словом Else разделитель (точка с запятой) не ставится.

Отметим, что большинство операторов в программах на языке Паскаль заканчиваются точкой с запятой, но после некоторых операторов точка с запятой не ставится. Сформулируем общие правила употребления точки с запятой:

1. Каждое описание переменной и определение константы заканчиваются точкой с запятой.

2. Каждый оператор в теле программы завершается точкой с запятой, если сразу за ним не следуют зарезервированные слова End, Else, Until.

3. После определенных зарезервированных слов, таких, как Then, Else, Var, Const, Begin, никогда не ставится точка с запятой.

Рассмотрим еще один пример.

Задача. Вывести на экран большее из двух данных чисел.

	Program Example1;
Var
  x, y : integer; {вводимые числа}
Begin
  writeln('Введите 2 числа '); {вводим два целых числа через пробел}
  readln(x,y);
  if x>y
    then
      writeln (x) {если х больше y, то выводим х}
    else
      writeln (y); {иначе выводим y}
    readln;
End.


Можно также использовать и сокращенную (неполную) форму записи условного оператора. Эта форма используется тогда, когда в случае невыполнения условия ничего делать не надо.

Неполная форма условного оператора имеет следующий вид.

	if <логическое выражение> 
  then
    <оператор>


Тогда если выражение, расположенное за служебным словом IF. в результате дает истину, выполняются действия после слова THEN, в противном случае эти действия пропускаются.

Задача. Составить программу, которая, если введенное число отрицательное, меняет его на противоположное.

	Program Chisla;
Var
  x : integer; {вводимое число}
Begin
  writeln('Введите число '); {вводим целое число}
  readln(x);
  if x<0
    then
      x:=-x;
  writeln (x);
  readln;
End.


Вложенные условные операторы

При решении задач часто приходится рассматривать не два, а большее количество вариантов. Это можно реализовать, используя несколько условных операторов. В этом случае после служебных слов Then и Else записывается новый условный оператор. Рассмотрим пример.

Задача. Вычислить значение функции:

Для решения этой задачи рассмотрим координатную прямую и отметим на ней промежутки, на которые разбиваются все значения переменной х.

Начнем записывать условный оператор:

если х>0
  то 
    вычислить у по формуле у=х-12
  иначе

Что же должно выполниться в случае "иначе"? На эту ветку оператора попадают все неположительные значения х. Если бы для этих чисел нужно было бы выполнить один и тот же оператор (или группу операторов), то проблемы бы не стояло. Но нам нужно этот промежуток разделить еще на две части (отрицательные и ноль) и для каждой части выполнить свой оператор. Поэтому ветка "иначе" будет содержать еще один условный оператор, и наш вложенный условный оператор будет иметь вид:

	если х>0
  то 
    у вычислить по формуле у=х-12
  иначе 
    если х=0
      то
        у вычислить по формуле у=5
      иначе
        у вычислить по формуле у=sqr(x);


Тогда фрагмент программы для решения этой задачи будет выглядеть так:

	if х>0
  then 
    у := х-12
  else 
    if х=0
      then 
        у := 5
      else 
        у := sqr(x);


Итак, когда оператор if появляется внутри другого оператора if, они считаются вложенными. Такое вложение используется для уменьшения числа необходимых проверок. Этот метод часто обеспечивает большую эффективность, однако одновременно он уменьшает наглядность программы. Не рекомендуется использовать более одного-двух уровней вложения if. За вторым уровнем вложения становится трудно восстановить последовательность проверки условий каждым условным оператором.

Если часть else используется во вложенных if, то каждое else соответствует тому if, которое ему непосредственно предшествует. Таким образом, при определении последовательности выполнения фрагментов нет двусмысленности.

Рассмотрите еще один пример.

Задача. Даны целые числа a, b, c. Если a ≤ b ≤ c, то все числа заменить их квадратами, если a>b>c, то каждое число заменить наибольшим из них, в противном случае сменить знак каждого числа.

Для решения этой задачи перепишем условие задачи следующим образом:

a:=a2, b:=b2, c:=c2, если a ≤ b ≤ c

b:=a, c:=a, если a>b>c

a:=-a, b:=-b, c:=-c, в остальных случаях

Программа для решения этой задачи представлена ниже.

	Program Example3;
Var
  a, b, c : integer; 
Begin
  writeln('Введите числа a, b, c'); 
  readln(a,b,c);
  if (a<=b) and (b<=c)
    then
      begin
        a:=sqr(a);
        b:=sqr(b);
        c:=sqr(c);
      end
    else
      if (a>b) and (b>c)
        then
          begin
            b:=a;
            c:=a;
          end
        else
          begin
            a:=-a;
            b:=-b;
            c:=-c;
          end;
  writeln(a,b,c);
  readln;
End.


Оператор выбора CASE

Ранее Вы познакомились с условным оператором If, который позволяет программе выполнять переходы на ту или иную ветвь по значению булева условия. Используя несколько операторов If, можно производить ветвление по последовательности условий. В приведенном фрагменте показано, как при помощи ряда операторов If можно преобразовать целое число (в диапазоне 0-9) к его словесному представлению:

	if Ziphra = 0
  then
    write ('Нуль');
if Ziphra = 1
  then
    write ('Единица');
if Ziphra = 2
  then
    write ('Два');


и т.д.

Вы уже, наверное, представили, насколько этот подход однообразный и утомительный. Язык Паскаль предоставляет для этих целей другую управляющую структуру (оператор выбора case), которая позволяет построить ветвление по ряду условий в форме, более удобной для чтения программ.

Оператор выбора позволяет выбрать одно из нескольких возможных продолжений программы. Параметром, по которому осуществляется выбор, служит так называемый ключ выбора (или селектор) - выражение любого типа (кроме типов REAL и STRING).

Общая форма записи следующая:

	case Выражение of
значение1 : оператор (группа операторов);
значение2 : оператор (группа операторов);
. . . . . . . . . . . . . . . . . . . . . .
значениеN : оператор (группа операторов)
else оператор (группа операторов);
end;


Оператор выбора работает следующим образом. Сначала вычисляется значение выражения, стоящего после зарезервированного слова case, а затем выполняется оператор (или составной оператор), соответствующий результату вычисления выражения.

Может случиться, что в списке выбора не окажется константы, равной вычисленному значению ключа. В этом случае управление передается оператору, стоящему за словом ELSE.

Например,

	case NUMBER mod 2 of
  0 : writeln (NUMBER, '- число четное')
else writeln (NUMBER, '- число нечетное');
end;


Если один оператор выполняется при нескольких значениях, то их можно перечислить через запятую.

	case MONTH of
  1, 2, 3 : writeln ('Первый квартал');
  4, 5, 6 : writeln ('Второй квартал');
  7, 8, 9 : writeln ('Третий квартал');
  10, 11, 12 : writeln ('Четвёртый квартал');
end;


Оператором может являться не только простой оператор, но также составной и пустой операторы.

	case CODE of
  1 : for i := 1 to 5 do
         writeln ('*******');
  2 : begin {составной оператор}
         x:=sqr(y-1); 
         writeln (x);
       end;
  3 : {пустой оператор}
       end;


Любому заданному значению селектора соответствует лишь один вход в списке операторов. Константы должны принадлежать тому же типу, что и селектор. Если селектор принимает значение, которому не соответствует ни один вход, то будет выполняться оператор, следующий за словом else. Если же этого оператора нет, то никакие альтернативы не будут выполняться.

Если оператор должен выполняться при нескольких последовательных значениях селектора, образующих некоторый промежуток, можно использовать следующую форму записи:

	case Chislo of
  0..9 : write ('Это число является цифрой');


Посмотрите, как применен оператор выбора при решении следующей задачи:

Задача. Написать программу преобразования цифр в слова.

	Program Number1;
Var
  a : integer; 
Begin
  writeln('Введите цифру '); 
  readln(a);
  if (a<0) or (a>9)
    then
      writeln ('Это число не является цифрой')
    else
      case a of
        0 : writeln ('ноль');
        1 : writeln ('один');
        2 : writeln ('два');
        3 : writeln ('три');
        4 : writeln ('четыре');
        5 : writeln ('пять');
        6 : writeln ('шесть');
        7 : writeln ('семь');
        8 : writeln ('восемь');
        9 : writeln ('девять');
      end;
  readln;
End.


	Program Number2;
Var
  a : integer; 
Begin
  writeln('Введите цифру '); 
  readln(a);
  case a of
    0 : writeln ('ноль');
    1 : writeln ('один');
    2 : writeln ('два');
    3 : writeln ('три');
    4 : writeln ('четыре');
    5 : writeln ('пять');
    6 : writeln ('шесть');
    7 : writeln ('семь');
    8 : writeln ('восемь');
    9 : writeln ('девять')
  else writeln ('Это число не является цифрой');
  end;
  readln;
End.


